unittest
{
	assert( [ "3", "1 2 3" ].parse.expand.solve == 1 );
	assert( [ "5", "3 11 14 5 13" ].parse.expand.solve == 2 );
	
	assert( [ "5", "2 2 2 2 2" ].parse.expand.solve == 2 );		// 追加ケース
}



import core.bitop;
import std.container;
import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	input.popFront;
	auto as = input.front.split.to!( long[] );
	return tuple( as );
}

auto solve( long[] as )
{
	// 2^^n-1 <= a < a+b == 2^^n <= 2a < 2^^n+1 なので、
	// a = max(as) のときの b は一意に決まるから、大きい順で貪欲にペアを作ればいい
	auto ans = 0L;
	auto rbt = redBlackTree!true( as );
	while( !rbt.empty )
	{
		auto a = rbt.back;
		rbt.removeBack;
		auto b = ( 2L << a.bsr ) - a;
		if( rbt.removeKey( b ) ) ans++;		// a に対応する b が見つかればペアにする
	}
	return ans;
}
